Eötvös Loránd Tudományegyetem
Természettudományi Kar

Számítógépes programok alkalmazása az analízisben

Szakdolgozat

Csillagvári Dániel
Matematika BSc, elemző szakirány

Témavezető:
Gémes Margit
Analízis Tanszék
Műszaki gazdasági tanár

Budapest
2012

Tartalom

Előszó

Szakdolgozatomban a többváltozós analízissel, ezen belül az $\mathbb{R}^p \to \mathbb{R}$ függvényekkel foglalkozom. Célom, hogy olyan ábrákat és eljárásokat készítsek a Maple program segítségével, melyek megkönnyítik a definíciók és tételek megértését, könnyebbé és gyorsabbá teszik a feladatmegoldást vagy azok ellenőrzését. Az általam készített eljárásokat részletesen leírom és elmagyarázom. Ezek segítséget nyújthatnak azoknak, akik most ismerkednek az anyagrésszel vagy már jól ismerik a feladatmegoldáshoz szükséges módszereket, és ezt minél gyorsabban szeretnék végezni számítógép segítségével. Az olvasóról feltételezem, hogy már használta a Maple programot, ezért nem célom a program alapvető utasításainak bemutatása, ismertetése.

Fejezetenként ismertetem a fontosabb definíciókat, tételeket, állításokat, majd példákon kersztül bemutatom a megoldási módszereket és azok számítógépes implementációját. A példákban kétváltozós függvények szerepelnek.

1. fejezet

$\mathbb{R}^p \to \mathbb{R}$ függvények

1.1. Parciális deriváltak

Ebben a fejezetben bevezetem a parciális deriváltak fogalmát és kimondom az ide vonatkozó tételeket, definíciókat. Ezek nagy segítségünkre lesznek a többváltozós függvények szélsőértékhelyeinek meghatározásában, valamint érintősíkok konstruálásában.

1.1. Definíció. *[1, 358. oldal].*
Ha egy többváltozós függvény néhány változóját rögzítjük és a függvényt a maradék, nem rögzített változók függvényeinek fogjuk fel, akkor az így kapott függvényeket az eredeti függvény *szekciófüggvényeinek* nevezzük.

Pirossal az $f(x,2) = x^2 + 4$, kékkel pedig az $f(2,y) = y^2 + 4$ szekciófüggvényt jelöltem.



1.1. ábra. $f(x,y) = x^2 + y^2$



1.2. ábra. $f(x,y) = x^2 + y^2$

1.2. Definíció. *[1, 359. oldal].*

Legyen az f függvény értelmezve az $a = (a_1, \ldots, a_p) \in \mathbb{R}^p$ pont egy környezetében. Rögzítsük az $a = (a_1, \ldots, a_p)$ pont koordinátáit az i-edik kivételével, és tekintsük a megfelelő

$$f_i(t) = f(a_1, \ldots, a_{i-1}, t, a_{i+1}, \ldots, a_p) \tag{1.1}$$

szekciófüggvényt. Az így kapott egyváltozós f_i függvény a_i pontban vett deriváltját (amennyiben létezik) az f függvény a pontban vett i-edik *parciális deriváltjának* nevezzük és $D_i f(a)$-val jelöljük.

Szemléltető példaként nézzük az

$$f(x, y) = \frac{1}{5} sin(x) \cdot x - 2y^2$$

függvényt és írjuk fel a $(3, 0)$ pontban vett parciális deriváltjait! A szekciófüggvények, amiket a következő oldalon található három ábrán kékkel, illetve pirossal jelöltem:

$$f(3, y) = \frac{3}{5} sin(3) - 2y^2$$

$$f(x, 0) = \frac{1}{5} sin(x) \cdot x$$

Ha az első egyváltozós függvényt deriváljuk, majd y -ba nullát helyettesítünk, akkor az eredeti f függvényünk $(3, 0)$ pontban vett y szerinti parciális deriváltját kapjuk, ami egyenlő 0 -val. A másik szekciófüggvénynél hasonlóan járunk el, így a $(3, 0)$ pontban vett parciális deriváltak:

$$f'_x(3, 0) = \frac{3}{5} cos(3) + \frac{1}{5} sin(3)$$

$$f'_y(3, 0) = -4y = 0$$

Az alábbi ábrákon láthatjuk az eredeti $f(x, y) = \frac{1}{5} sin(x) \cdot x - 2y^2$ függvényt, valamint annak szekciófüggvényeit kétdimenzós koordinátarendszerben. Tehát, ha az $f(x, y)$ függvény valamelyik parciális deriváltját szeretnénk meghatározni, akkor a megfelelő egyváltozós függvény deriváltját keressük.

Ha szekciófüggvényeket szeretnénk Maple-ben ábrázolni, akkor ajánlanám az általam írt néhány sort, amiket az ábrák után találunk. Paraméterként az f függvényt, az (a, b) pontot, valamint azt a tarományt kell megadni, ahol látni szeretnénk a függvényeket. A *plot3d* parancs az $f(x, y)$, a két *spacecurve* pedig ennek szekciófüggvényeit rajzolja ki, majd a *display* segítségével ugyanabban a koordinátarendszerben jelenítjük meg őket. A *spacecurve* görbék ábrázolására szolgál, paraméterként az (x, y, z) pontokhoz tartozó függvényeket kell megadnunk, a kódban ezeket a szögletes zárójel között találjuk. Bővebb információ a görbékről: [1, 104. oldal].



1.3. ábra. $f(x, y) = \frac{1}{5}sin(x) \cdot x - 2y^2$



1.4. ábra. $f(3, y) = -\frac{3}{5}sin(3) - 2y^2$



1.5. ábra. $f(x, 0) = \frac{1}{5}sin(x) \cdot x$

```
with(plots):
szekcio:=proc(f,a,b,c,d,e,g)
local d13,d13px,d13py:
plot3d(f,x=c..d,y=e..g,color="Silver",axes=boxed,numpoints=1000);d13:=%:
spacecurve([x,b,subs(y=b,f)],x=c..d,color=red,axes=boxed,thickness=3);d13px:=%:
spacecurve([a,y,subs(x=a,f)],y=e..g,color=blue,axes=boxed,thickness=3);d13py:=%:
display(d13,d13px,d13py)
end proc:
#szekcio(f,a,b, a tégla határai: x1=c,x2=d,y1=e,y2=g)
```

Ennek tükrében a fenti *1.3. ábra* a következő parancs beírásával készült:

```
szekcio(sin(x)*x/5-2*y^2, 3, 0, 0, 12, -2, 2);
```

1.3. Definíció. *[3].*
Ha a p-változós f függvény i-edik parciális deriváltjának létezik a j-edik parciális deriváltja a-ban, akkor ezt az f függvény *a-beli ji-edik másodrendű parciális deriváltjának* nevezzük és $D_{ji}f(a)$-val vagy $f''_{ij}(a)$-val jelöljük.

1.1. Tétel (Young-tétel). *[3].*
Ha a p-változós $f(x_1, x_2, \ldots, x_p)$ függvény $D_i f$ és $D_j f$ parciális deriváltjai léteznek egy $a \in \mathbb{R}^p$ pont egy kötnyezetében és a-ban differenciálhatóak, akkor $D_{ij}f(a) = D_{ji}f(a)$.
Bizonyítás nélkül.

Példának vegyük az $f(x, y) = x^2 \cdot 2^y$ függvényt! Az elsőrendű parciális deriváltak:

$$f'_x(x, y) = 2x \cdot 2^y$$

$$f'_y(x, y) = x^2 \cdot (log2) \cdot 2^y$$

Ha az elsőrendű parciális deriváltakat ismét deriváljuk, de most a függvény másik változója szerint, akkor a *Young-tétel* alapján azt kapjuk, hogy

$$f''_{xy}(x, y) = 2x \cdot (log2) \cdot 2^y = f''_{yx}(x, y)$$

Maple-ben rendkívül egyszerűen állíthatjuk elő a parciális deriváltakat, mindössze az f függvényt és azt a változót kell megadni, ami szerint deriválni szeretnénk. Például az előző függvény első és másodrendű parciális deriváltjai a következő sorok beírásával kaphatók:

```
diff(x^2*2^y,x);
diff(x^2*2^y,y);
diff(diff(x^2*2^y,x),y);
```

1.2. Érintősík

1.4. Definíció. *[1, 377. oldal].*

Legyen $(a, b) \in \mathbb{R}^2$ rögzített, és legyen f értelmezve az (a, b) pont egy környezetében. Azt mondjuk, hogy az S sík a graph f grafikon *érintősíkja* az $(a, b, f(a, b))$ pontban, ha S átmegy az $(a, b, f(a, b))$ ponton, és S egy olyan g elsőfokú polinom grafikonja, amelyre teljesül, hogy

$$\lim_{(x,y)\to(a,b)} \frac{f(x, y) - g(x, y)}{|(x, y) - (a, b)|} = 0$$

Az f függvény grafikonjának akkor és csak akkor van érintősíkja az $(a, b, f(a, b))$ pontban, ha f differenciálható (a, b)-ben. Ekkor az érintősík egyenlete:

$$z = D_1 f(a, b)(x - a) + D_2 f(a, b)(y - b) + f(a, b) \tag{1.2}$$

Adjuk meg az $f(x, y) = -x^2 - y^2$ fügvény érintősíkjának az egyenletét az $(1, -1)$ pontban! Ehhez először elő kell állítanunk a parciális deriváltakat, amelyek (a, b)-ben:

$$D_1 f(a, b) = -2$$

$$D_2 f(a, b) = 2$$

Ez alapján az értintősík egyenlete:

$$z = -2(x - 1) + 2(y + 1) - 2 = -2x + 2y + 2$$

Maple-ben érintősík ábrázolására a következő sorokat alkalmaztam:

```
with(plots):
erintosik := proc(f,a,b,c,d,e,g)
local fv,erint,sp1,sp2,sp3,sp4;
plot3d(f(x,y),x=c..d,y=e..g,color="Silver",transparency=0);fv:=%:
plot3d(subs(x=a,y=b,f(x,y))+evalf(subs(x=a,diff(f,x)))*(x-a)+
  evalf(subs(y=b,diff(f,y)))*(y-b),x=a-(1/4)*(d-c)..a+(1/4)*(d-c),
  y=b-(1/4)*(g-e)..b+(1/4)*(g-e),color="Yellow",transparency=0.1);erint:=%:
spacecurve([x,b,subs(y=b,f(x,y))],x=c..d,color=red,axes=boxed,thickness=3);sp1:=%:
spacecurve([a,y,subs(x=a,f(x,y))],y=e..g,color=blue,axes=boxed,thickness=3);sp2:=%:
spacecurve([x,b,-8],x=c..d,color=red,axes=boxed,thickness=1);sp3:=%:
spacecurve([a,y,-8],y=e..g,color=blue,axes=boxed,thickness=1);sp4:=%:
display(fv,erint,sp1,sp2,sp3,sp4);
end proc:
```

A következő parancs az alábbi ábrán látható érintősíkot állítja elő:

```
erintosik(-x^2-y^2, 1, -1, -2, 2, -2, 2);
```



1.6. ábra.

1.3. Iránymenti derivált és gradiens

1.5. Definíció. *[1, 378. oldal].*

Legyen $v \in \mathbb{R}^p$ egy egységvektor. A $t \to f(a+tv)$ függvény 0 pontbeli deriváltját (ha létezik) az f függvény a pontbeli v irányú *iránymenti deriváltjának* nevezzük, és $D_v f(a)$-val jelöljük. Tehát feltéve, hogy a limesz létezik és véges:

$$D_v f(a) = \lim_{t \to 0} \frac{f(a+tv) - f(a)}{t}$$

1.2. Tétel. *[1, 378. oldal].*

Ha az f függvény differenciálható az $a \in \mathbb{R}^p$ pontban, akkor minden $v \in \mathbb{R}^p$ vektorra a $t \to f(a+tv)$ egyváltozós függvény differenciálható a 0 pontban, és a deriváltja $\langle f'(a), v \rangle$. Speciálisan, ha $|v| = 1$, akkor a $D_v f(a)$ iránymenti derivált létezik és az értéke $D_v f(a) = \langle f'(a), v \rangle$. *Bizonyítás nélkül.*

Ha például egy kétváltozós $f(x,y)$ függvény a pontbeli v irányú iránymenti deriváltjáról beszélünk, akkor annak az egyváltozós függvénynek a deriváltjára gondolunk, amelyiket az eredeti f függvény grafikonjának és az a, valamint az $a+v$ pontokon átmenő, az x és y koordinátatengelyek által kifeszített síkra merőleges sík metszeteként kapunk.

Ha azt a feladatot kapnánk, hogy számoljuk ki az

$$f(x,y) = -x^4 - y^2$$

függvény $a=(-1,-1)$ pontbeli $v=(1,1)$ irányú iránymenti deriváltját, akkor az alábbi ábrán pirossal jelölt egyváltozós függvény deriváltját szeretnénk kiszámolni a zölddel jelölt pontban. A derivált értékének kiszámításához a v vektort normalizáljuk, így

$$v = (\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}})$$

$$f'(x,y) = (-4x^3, -2y)$$

$$f'(a) = (4,2)$$

Ebből:

$$D_v f(a) = \langle f'(a), v\rangle = \langle (4,2), \left(\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right)\rangle = \frac{6}{\sqrt{2}}.$$



1.7. ábra.

A fenti képet a következő Maple kóddal készítettem:

```
irany := proc(f,a,b,c,d,v1,v2)
local fv,irany,sp1,sp2,sp3,v;
plot3d(f(x,y),x=-c..c,y=-d..d,color="Silver",transparency=0);fv:= %:
spacecurve([a+t,b+t,subs(x=a+t,y=b+t,f(x,y))],t=-1..5,color=red,axes=boxed,
  thickness=3);sp1:=%:
spacecurve([a,b,t],t=-20..0,color=green,axes=boxed,thickness=3);sp2:=%:
spacecurve([a+t,b+t,-20],t=-1..3,color=red,axes=boxed,thickness=1);sp3:=%:
v := arrow(<-1,-1,-20>,<1,1,0>,shape=cylindrical_arrow,color=blue);
display(fv,sp1,sp2,sp3,v, view=[-c..c,-d..d,-20..0]);
end proc:
irany(-x^4-y^2, -1, -1, 2, 2, 1, 1);
```

1.1. Megjegyzés. *[1, 379. oldal].*
Tegyük fel, hogy a $D_i f(a)$ parciális deriváltak nem mind nullák, azaz hogy az $f'(a)$ deriváltvektor nem mindegyik koordinátája nulla. Ha $|v| = 1$, akkor $\langle f'(a), v\rangle = |f'(a)| \cdot cos\alpha$, ahol α az $f'(a)$ és v vektor által bezárt szög. Így $\langle f'(a), v\rangle \leq |f'(a)|$, és egyenlőség csak akkor áll, ha v az $f'(a)$ irányába mutat. Más szóval az f függvény grafikonján végzett „hegymászás” az $f'(a)$ vektor irányába a legmeredekebb. Erre a tényre utalva az $f'(a)$ deriváltvektort *gradiensnek* is szokás nevezni.

Például az előző feladatban szereplő $f(x, y) = -x^4 - y^2$ függvény deriváltvektora az $a = (-1, -1)$ pontban $(4, 2)$, így a „hegymászás” a $v = (2, 1)$ irányba a legmeredekebb.

1.4. Lokális szélsőértékek és nyeregpontok

A most következő fejezetekben megismerkedünk a lokális szélsőérték és a nyeregpont fogalmával, majd ezek meghatározására koncentrálunk.

1.6. Definíció. *[1, 366. oldal].*
Azt mondjuk, hogy az f függvénynek az $a \in \mathbb{R}^p$ pontban *lokális minimuma* (illetve *lokális maximuma)* van, ha a-nak van olyan U környezete, amelyben f értelmezve van, és minden $x \in U$-ra $f(x) \geq f(a)$ $(f(x) \leq f(a))$. Ekkor az a pontot az f függvény *lokális minimumhelyének* (illetve *lokális maximumhelyének)* nevezzük.
Ha minden $x \in U \backslash \{a\}$ pontra $f(x) < f(a)$ (illetve $f(x) > f(a)$), akkor *szigorú lokális maximumról és maximumhelyről* (illetve *minimumról és minimumhelyről)* beszélünk.
A lokális maximumot és minimumot közösen *lokális szélsőértéknek*, a lokális maximumhelyet és minimumhelyet közösen *lokális szélsőértékhelynek* nevezzük.

1.3. Tétel. *[1, 367. oldal].*
Ha az f függvénynek lokális szélsőértéke van az $a \in \mathbb{R}^p$ pontban, és f-nek léteznek a parciális deriváltjai a-ban, akkor $D_i f(a) = 0$ minden $i = 1, \ldots, p$-re.

Ez a tétel tulajdonképpen azt mondja ki, hogy ha egy többváltozós függvénynek az a pontban lokális szélsőértéke van, akkor a-ban az összes szekciófüggvényének deriváltja nulla. A megfordítás nem igaz, mert előfordulhat, hogy az egyik szekciófüggvénynek lokális minimuma, míg a másiknak lokális maximuma van az adott pontban. Ebben az esetben mondjuk azt, hogy a függvénynek *nyeregpontja* van a-ban. Ezen felül létezik olyan függvény, aminek a parciális deriváltjai mind nullák $(0,0)$-ban, az összes szekciófüggvényének szigorú lokális minimuma van $(0,0)$-ban, de magának a függvénynek nincs lokális szélsőértékhelye $(0,0)$-ban. A későbbiekben egy feladat keretében találkozhatunk ilyen függvénnyel. A baloldali képen az $f(x,y) = -x^2 - y^2$ függvényt és annak szekciófüggvényeit látjuk, $(0,0)$-ban szidorú lokális maximum van. A jobboldali képen az $f(x,y) = -x^2 + y^2$ függvényt és annak szekciófüggvényeit látjuk, $(0,0)$-ban nyeregpontja van.



1.8. ábra. *szigorú lokális maximum*



1.9. ábra. *nyeregpont*

1.4. Tétel. *[1, 367. oldal].*

Legyen $A \subset \mathbb{R}^p$ korlátos és zárt, legyen $f : A \to \mathbb{R}$ folytonos, és tegyük fel, hogy f-nek léteznek a parciális deriváltjai A belsejének minden pontjában. Ekkor f a legnagyobb (legkisebb) értékét vagy A határán veszi fel, vagy pedig egy olyan a belső pontban, ahol $D_i f(a) = 0$ minden $i = 1, \ldots, p$-re. *Bizonyítás nélkül.*

1.5. Többváltozós függvények közelítése másodfokú polinommal

A kérdés, hogy hogyan tudnánk jól közelíteni egy n-változós függvényt egy adott a pont közelében. Például az érintő hipersíkkal ($n = 2$ esetén érintősíkkal), aminek az egyenlete:

$$z = f(a) + f'_{x_1}(a)(x_1 - a_1) + f'_{x_2}(a)(x_2 - 1_2) + \cdots + f'_{x_n}(a)(x_n - a_n). \tag{1.3}$$

Azzal a céllal szeretnénk közelíteni a függvényt, hogy eldönthessük, az adott pontban lokális minimuma vagy lokális maximuma van-e. Erre nem igazán alkalmas az érintő hipersík, mert ha a parciális deriváltak mind nullák, akkor az egyenletből csak $z = f(a)$ marad meg, a többi tag nulla. Az ötlet az, hogy közelítsük másodfokú polinommal. Egy általános n-változós másodfokú polinom a következőképpen írható fel:

$$p(x_1, x_2, \ldots, x_n) = b + c_1x_1 + \cdots + c_nx_n + d_{11}x_1^2 + d_{12}x_1x_2 + \cdots + d_{1n}x_1x_n +$$

$$+ d_{21}x_2x_1 + d_{22}x_2^2 + d_{2n}x_2x_n + \cdots + d_{nn}x_n^2$$

Először legyen $a = (0, \ldots, 0)$, tehát keressük az f-et az origóban legjobban közelítő másodfokú p polinomot. Az a polinom fogja jól közelíteni, amelynek ugyanazok nullában a nulladik, első és második parciális deriváltjai.

- *Nulladik:*

$$f(0) = p(0) = b \Longrightarrow b = f(0)$$

- *Első:*

$$f'_{x_i}(0) = p'_{x_i}(0) = c_i \Longrightarrow c_i = f'_{x_i}(0)$$

  Tehát $p'_{x_i} = c_i$+valami, ami 0-ban nulla.

- *Második:*
  - *Ha $i = j$:*

$$f''_{x_ix_i}(0) = p''_{x_ix_i}(0) = 2d_{ii}$$

  - *Ha $i \neq j$:*

$$f''_{x_ix_j}(0) = p''_{x_ix_j}(0) = d_{ij} + d_{ji}$$

  Tehát, ha $d_{ij} = \frac{1}{2}f''_{x_ix_j}(0)$, akkor $f''_{x_ix_j}(0) = p''_{x_ix_j}(0)$ $\forall i, j$-re.

Tehát az f függvényt (remélhetőleg) jól tudjuk közelíteni $(0, \ldots, 0)$ körül az alábbi másodfokú polinommal:

$$f(0) + f'_{x_1}(0) \cdot x_1 + \cdots + f'_{x_n}(0) \cdot x_n + \frac{1}{2}f''_{x_1x_1}(0) \cdot x_1^2 + \frac{1}{2}f''_{x_1x_2}(0) \cdot x_1x_2 + \cdots + \frac{1}{2}f''_{x_nx_n}(0) \cdot x_n^2$$

Más alakban:

$$f(0) + \sum_{i=1}^{n} f'_{x_i}(0) \cdot x_i + \sum_{i=1}^{n} \sum_{j=1}^{n} \frac{1}{2} f''_{x_i x_j}(0) \cdot x_i x_j \tag{1.4}$$

Ez a polinom tényleg jól közelíti, ha f kétszer differenciálható nullában. *Bizonyítás nélkül.*

1.7. Definíció. Ez az f *második Taylor-polinomja* $(0, \dots, 0)$-ban.
Jelölés: $t_2(x_1, \dots, x_n)$.

Tegyük fel, hogy f-nek $(0, \dots, 0)$-ban lokális minimium/maximum jelöltje van, azaz $f'_{x_1}(0) = \dots = f'_{x_n}(0) = 0$. Ekkor

$$t_2(x_1, \dots, x_n) = f(0) + \sum_{i=1}^{n} \sum_{j=1}^{n} \frac{1}{2} f''_{x_i x_j}(0) \cdot x_i x_j. \tag{1.5}$$

Mivel t_2 jól közelíti f-et 0 közelében, lokális minimum és maximum szempontjából úgy viselkedik, mint f, tehát vizsgálhatjuk azt, hogy t_2-nek mikor van lokális minimuma, illetve maximuma nullában. Ekkor a kérdés az, hogy egy nulla körüli gömbben

$$t_2(0, \dots, 0) \leq t_2(x_1, \dots, x_n)$$

teljesül vagy sem. Ha igen, akkor t_2-nek lokális minimuma van nullában. Mivel $t_2(0, \dots, 0) = f(0, \dots, 0)$, elég azt vizsgálni, hogy

$$0 \leq \sum_{i=1}^{n} \sum_{j=1}^{n} \frac{1}{2} f''_{x_i x_j}(0) \cdot x_i x_j \tag{1.6}$$

teljesül-e.

1.6. Kvadratikus alak

1.8. Definíció. A *kvadratikus alak* olyan n-változós polinom, amelyben minden tag másodfokú, azaz

$$q(x_1, \dots, x_n) = \sum_{i=1}^{n} \sum_{j=0}^{n} c_{ij} x_i x_j$$

alakú.

Tehát az (1.6) egyenlőtlenség jobb oldalán lévő kifejezés kvadratikus alak.
A q kvadratikus alak:

- *pozitív szemidefinit*, ha $q(x) \geq 0 \qquad \forall x \in \mathbb{R}^n$
- *negatív szemidefinit*, ha $q(x) \leq 0 \qquad \forall x \in \mathbb{R}^n$

- *pozitív definit*, ha $q(x) > 0 \qquad \forall x \neq (0, \dots, 0) \in \mathbb{R}^n$
- *negatív definit*, ha $q(x) < 0 \qquad \forall x \neq (0, \dots, 0) \in \mathbb{R}^n$
- *indefinit*, ha fölvesz pozitív és negatív értéket is.

Tehát t_2-nek a következő szélsőértékei lehetnek 0-ban, ha

$$\sum_{i=1}^{n} \sum_{j=1}^{n} \frac{1}{2} f''_{x_i x_j}(0) \cdot x_i x_j$$

- *pozitív szemidefinit:* - lokális minimum
- *pozitív definit:* - szigorú lokális minimum
- *negatív szemidefinit:* - lokális maximum
- *negatív definit:* - szigorú lokális maximum
- *indefinit:* - egyik sem (nyeregpont)

Legyen C az a mátrix, ahol $c_{ij} = f''_{x_i x_j}(0)$. A C mátrix pozitív definit, ha a mátrixban minden bal felső főminor determinánsa pozitív.

$$\begin{vmatrix} c_{11} \end{vmatrix} > 0; \qquad \begin{vmatrix} c_{11} & c_{12} \\ c_{21} & c_{22} \end{vmatrix} > 0; \qquad \dots; \qquad \begin{vmatrix} c_{11} & \dots & c_{1n} \\ \vdots & \ddots & \vdots \\ c_{n1} & \dots & c_{nn} \end{vmatrix} > 0$$

A C mátrix negatív definit, ha a mátrixban a bal felső főminorok determinánsa felváltva negatív és pozitív.

$$\begin{vmatrix} c_{11} \end{vmatrix} < 0; \qquad \begin{vmatrix} c_{11} & c_{12} \\ c_{21} & c_{22} \end{vmatrix} > 0; \qquad \begin{vmatrix} c_{11} & c_{12} & c_{13} \\ c_{21} & c_{22} & c_{23} \\ c_{31} & c_{32} & c_{33} \end{vmatrix} < 0; \qquad \dots$$

1.5. Tétel. *[3].*
Tegyük fel, hogy f kétszer differenciálható $a \in \mathbb{R}^n$-ben. Ekkor, ha

1. f-nek a-ban *lokális minimuma/maximuma* van, akkor $D_i f(a) = 0 \qquad \forall i$-re és a

$$\sum_{i=1}^{n} \sum_{j=1}^{n} D_{ji} f(a) \cdot h_i h_j$$

kvadratikus alak *pozitív/negatív szemidefinit.* (Szükséges feltétel)

2. Ha $D_i f(a) = 0 \qquad \forall i$-re és a

$$\sum_{i=1}^{n}\sum_{j=1}^{n} D_{ji} f(a) \cdot h_i h_j$$

kvadratikus alak *pozitív/negatív definit*, akkor f-nek a-ban *szigorú lokális minimuma/maximuma* van. (Elégséges feltétel)
A megfordítások nem igazak.

1.1. Következmény.

$$\sum_{i=1}^{n}\sum_{j=1}^{n} D_{ji} f(a) \cdot h_i h_j \quad \text{indefinit}$$

$$\Downarrow$$

f-nek nincs a-ban szélsőértéke, hanem *nyeregpontja* van.

Az előző $(D_{ij} f(a))$ mátrix neve *Hesse-mátrix.*

A fenti tételben és következményben azért szerepel $x_i x_j$ helyett $h_i h_j$, mert az a-beli 2. Taylor-polinomban nem x_i, hanem $(x_i - a_i)$ szerepel.

$$f(a) + \sum_{i=1}^{n} f'_{x_i}(a) \cdot (x_i - a_i) + \sum_{i=1}^{n}\sum_{j=1}^{n} \frac{1}{2} f''_{x_i x_j}(a) \cdot (x_i - a_i)(x_j - a_j) \tag{1.7}$$

Ezzel befejeztük a szélsőértékhelyek megtalálásához szükséges definíciók, tételek tárgyalását. A következő fejezetben példákon keresztül ismerhetjük meg ezek alkalmazását.

2. fejezet

Feladatok

2.1. Feladat. *Vizsgáljuk az $f(x,y) = x^4 + y^4 + 4xy + 2$ függvény lokális szélsőértékhelyeit!*

Lokális szélsőértékhelyek ott lehetnek, ahol a parciális deriváltak nullák.

$$f'_x = 4x^3 + 4y = 0 \tag{2.1}$$

$$f'_y = 4y^3 + 4x = 0 \tag{2.2}$$

Az első egyenletből $-x^3 = y$ -t kapunk, a másodikból $-y^3 = x$ -et. Ebből a lehetséges szélsőértékhelyek a $(-1,1), (0,0), (1,-1)$ pontban vannak. Most írjuk fel a másodrendű parciális deriváltakat!

$$f''_{xx} = 12x^2; \qquad f''_{yy} = 12y^2; \qquad f''_{xy} = f''_{yx} = 4$$

Számoljuk ki értéküket ezekben a pontokban és írjuk föl a hozzájuk tartozó Hesse-mátrixot!

$$f''_{xx}(-1,1) = 12; \qquad f''_{yy}(-1,1) = 12; \qquad f''_{xy}(-1,1) = f''_{yx}(-1,1) = 4; \qquad \begin{vmatrix} 12 & 4 \\ 4 & 12 \end{vmatrix}$$

$$f''_{xx}(0,0) = 0; \qquad f''_{yy}(0,0) = 0; \qquad f''_{xy}(0,0) = f''_{yx}(0,0) = 4; \qquad \begin{vmatrix} 0 & 4 \\ 4 & 0 \end{vmatrix}$$

$$f''_{xx}(1,-1) = 12; \qquad f''_{yy}(1,-1) = 12; \qquad f''_{xy}(1,-1) = f''_{yx}(1,-1) = 4; \qquad \begin{vmatrix} 12 & 4 \\ 4 & 12 \end{vmatrix}$$

Mivel $(1,-1)$-ben és $(-1,1)$-ben a Hesse-mátrix pozitív definit, ezekben a pontokban az f függvénynek szigorú lokális minimuma van, míg $(0,0)$-ban nincs lokális szélsőértéke, mert a Hesse-mátrix indefinit.

2.2. Feladat. *Vizsgáljuk az $f(x,y) = x^4 + x^4$ függvény lokális szélsőértékhelyeit!*

Ezt a feladatot tulajdonképpen ránézésre meg lehet oldani, de mi lenne, ha a tanult módszerrel próbálkoznánk? Tehát felírjuk az első- és másodrendű parciális deriváltakat és a lehetséges pontokhoz tartozó Hesse-mátrixot.

$$f'_x = 4x^3; \qquad f'_y = 4y^3$$

Mindkettő egyszerre akkor nulla, ha $(x,y) = (0,0)$.

$$f''_{xx} = 12x^2; \qquad f''_{yy} = 12y^2; \qquad f''_{xy} = f''_{yx} = 0$$

$$f''_{xx}(0,0) = 0; \qquad f''_{yy}(0,0) = 0; \qquad f''_{xy}(0,0) = f''_{yx}(0,0) = 0; \qquad \begin{vmatrix} 0 & 0 \\ 0 & 0 \end{vmatrix}$$

Azt látjuk, hogy a Hesse-mátrix alapján nem tudunk dönteni, a kvadratikus alak felírásával ugyanez a helyzet. Mivel a feladat az volt, hogy az origóban vizsgáljuk a szélsőértéket, megnézzük, hogy a függvény ebben a pontban milyen értéket vesz föl. Ebben az esetben ez adott, itt az értéke nulla. Az origó kivételével a koordinátatengelyeken is mindenhol, valamint mind a négy síknegyedben is mindenhol pozitív a függvény értéke. Ebből következik, hogy az $f(x,y) = x^4 + x^4$ függvények az origóban szigorú lokális minimuma van.

2.3. Feladat. *Vizsgáljuk a következő függvény szélsőértékét az origóban!*

$$f(x,y) = \begin{cases} \dfrac{xy}{x^2+y^2} & , ha (x,y) \neq (0,0) \\ 0 & , ha (x,y) = (0,0) \end{cases}$$

Ez a ránézésre kicsit nehezebb feladat is megoldható az előző ötlet alapján. Mindkét koordinátatengely mentén a függvény értéke nulla, a szomszédos síknegyedekben pedig ellentétes előjelű, amiből az következik, hogy hogy az f függvénynek nincs lokális szélsőértékhelye az origóban.

2.4. Feladat. *Vizsgáljuk az $f(x,y) = (y-x^2)(y-3x^2)$ függvény lokális szélsőértékhelyeit!*

Lokális szélsőértékhelyek ott lehetnek, ahol a parciális deriváltak nullák.

$$f'_x = 12x^3 - 8xy = 0 \tag{2.3}$$

$$f'_y = 2y - 4x^2 = 0 \tag{2.4}$$

A második egyenletből $y = 2x^2$ -et kapunk, amit az elsőbe behelyettesítve $12x^3 = 16x^3$ adódik, ami csak úgy lehetséges, ha $x = 0$. Ezt a második egyenletbe behelyettesítve $y = 0$ -át kapunk. Ebből következik, hogy az f függvénynek csak a $(0,0)$ pontban lehet lokális szélsőértékhelye. Ennek megállapításához először írjuk fel a másodrendű parciális deriváltakat!

$$f''_{xx} = 36x^2 - 8y; \qquad f''_{yy} = 2; \qquad f''_{xy} = f''_{yx} = -8x$$

Most számoljuk ki az értéküket a (0,0) pontban!

$$f''_{xx}(0,0) = 0; \qquad f''_{yy}(0,0) = 2; \qquad f''_{xy}(0,0) = f''_{yx}(0,0) = 0$$

Most már fel tudjuk írni a (0,0) ponthoz tartozó Hesse-mátrixot, ez a következő:

$$\begin{vmatrix} 0 & 0 \\ 0 & 2 \end{vmatrix}$$

Mivel ez a mátrix a tanult tétel alapján nem pozitív definit, nem negatív definit és nem indefinit, arra a megállapításra jutottunk, hogy az f függvénynek a (0,0) pontban nincs szigorú lokális szélsőértékhelye, de lehet lokális minimuma vagy maximuma. A kvadratikus alak: $q = 2y^2$, ami pozitív szemidefinit. Azért nem pozitív definit, mert akkor nulla, ha $y = 0$, miközben x bármennyi lehet. Ebből következik, hogy f -nek (0,0)-ban lokális minimuma lehet.

Vegyük észre, hogy $f(x,y)$ értéke akkor nulla, ha $y = x^2$ vagy $y = 3x^2$! Az összes többi (x,y) pontban az f függvény vagy pozitív vagy negatív értékeket vesz föl. Legyen $g(x) = x^2$, valamint $h(x) = 3x^2$. Mindkét függvénynek szigorú lokális minimuma van a nulla pontban, itt az értékük nulla, ez az egyetlen közös pontjuk, értelmezési tartományuk a teljes számegyenes és folytonosak. Ebből következik, hogy az (x, y) síkot négy tartományra osztják. Az f függvény pedig egy tartományon belül csak azonos előjelű értékeket vehet föl, különben máshol is nulla lenne. Nézzük meg, hogy f negatív vagy pozitív a g és h közötti tarományban! Legyen például $y = 2x^2$. Ezt behelyettesítve f-be, a következőt kapjuk:

$$f(x, 2x^2) = 4x^4 + 3x^4 - 8x^4 = -x^4$$

Tehát, ha $g(x) < y < h(x)$, akkor $f(x,y) < 0$.
Legyen most például $y = 4x^2$. Ezt behelyettesítve f-be, a következőt kapjuk:

$$f(x, 2x^2) = 16x^4 + 3x^4 - 16x^4 = 3x^4$$

Tehát, ha $h(x) < y$, akkor $f(x,y) > 0$.
Ebből már következik, hogy $f(x,y)$ a nulla pont bármely környezetében pozitív és negatív értékeket is felvesz, tehát az origóban nincs lokális szélsőértékhelye.

2.5. Feladat. *Vizsgáljuk a következő függvény lokális szélsőértékhelyeit!*

$$f(x,y) = \begin{cases} (y - e^{-\frac{1}{x^2}})(y - 3e^{-\frac{1}{x^2}}) & , ha (x,y) \neq (0,0) \\ 0 & , ha (x,y) = (0,0) \end{cases}$$

A megoldás ötlete megegyezik az előző feladatnál alkalmazottal. Ezalatt az (x, y) síkot négy tartományra osztó két egyváltozós függvényre gondolok. A különbség annyi, hogy $e^{-\frac{1}{x^2}}$ nullában nincs értelmezve, ezért a két függvényt f-hez hasonlóan esetszétválasztással kell megadni. A megoldás ugyanaz, tehát nincs lokális szélsőértékhely a $(0, 0)$ pontban, bár a *2.2. ábra* nagyon megtévesztő.

Tehát $f(x,y)$ értéke akkor nulla, ha $y = e^{-\frac{1}{x^2}}$ vagy $y = 3e^{-\frac{1}{x^2}}$ vagy $y = 0$.

Az összes többi (x,y) pontban az f függvény vagy pozitív vagy negatív értékeket vesz föl. Legyen

$$g(x) = \begin{cases} e^{-\frac{1}{x^2}} & ,x \neq 0 \\ 0 & ,x = 0 \end{cases}$$

$$h(x) = \begin{cases} 3e^{-\frac{1}{x^2}} & ,x \neq 0 \\ 0 & ,x = 0 \end{cases}$$

Mindkét függvénynek szigorú lokális minimuma van a nulla pontban, itt az értékük nulla, ez az egyetlen közös pontjuk, értelmezési tartományuk a teljes számegyenes és folytonosak. Ebből következik, hogy az (x,y) síkot négy tartományra osztják. Az f függvény pedig egy tartományon belül csak azonos előjelű értékeket vehet föl, különben máshol is nulla lenne.

Nézzük meg, hogy f negatív vagy pozitív a g és h közötti tarományban! Legyen például

$$y = \begin{cases} 2e^{-\frac{1}{x^2}} & ,x \neq 0 \\ 0 & ,x = 0 \end{cases}$$

Ezt behelyettesítve f-be, a következőt kapjuk:

$$f(x, 2e^{\frac{-1}{x^2}}) = \begin{cases} (2e^{-\frac{1}{x^2}} - e^{-\frac{1}{x^2}})(2e^{-\frac{1}{x^2}} - 3e^{-\frac{1}{x^2}}) & ,(x,y) \neq (0,0) \\ 0 & ,(x,y) = (0,0) \end{cases}$$

Tehát, ha $g(x) < y < h(x)$, akkor $f(x,y) < 0$.

Legyen most például

$$y = \begin{cases} 4e^{-\frac{1}{x^2}} & ,x \neq 0 \\ 0 & ,x = 0 \end{cases}$$

Ezt behelyettesítve f-be, a következőt kapjuk:

$$f(x, 4e^{-\frac{1}{x^2}}) = \begin{cases} (4e^{-\frac{1}{x^2}} - e^{-\frac{1}{x^2}})(4e^{-\frac{1}{x^2}} - 3e^{-\frac{1}{x^2}}) & ,(x,y) \neq (0,0) \\ 0 & ,(x,y) = (0,0) \end{cases}$$

Ebből látszik, ha $h(x) < y$, akkor $f(x,y) > 0$.

Ebből már következik, hogy $f(x,y)$ a nulla pont bármely környezetében pozitív és negatív értékeket is felvesz, tehát az origóban nincs lokális szélsőértékhelye. Az alábbi két képet a feladatokhoz készítettem.



2.1. ábra. *2.4. feladat*



2.2. ábra. *2.5. feladat*

Mindkét példa jól szemlélteti, hogy szélsőértékhelyek keresésénél miért nem elég a parciális deriváltakat vizsgálni, valóban érdekes feladatok. A *2.2* ábra kójda:

```
g:=piecewise(0<>x,(y-exp(-1/x^2))*(y-3*exp(-1/x^2)),y^2):
plot3d(g,x=-1..1,y=-1..1,scaling=constrained,color="Silver",
  numpoints=5000, transparency=0):g1:=%:
plot3d(0,x=-1..1,y=-1..1,scaling=constrained,color="Gold"):g2:=%:
spacecurve([t,exp(-1/t^2),0],t=-1..1,color=red,axes=boxed,thickness=3):g3:=%:
spacecurve([t,3*exp(-1/t^2),0],t=-1..1,color=red,axes=boxed,thickness=3):g4:=%:
display(g1,g2,g3,g4,view=[-1..1,-1..1,-1..1]);
```

3. fejezet

Maple eljárás szélsőértékhelyek meghatározására

Ebben a fejezetben az általam írt Maple eljárást mutatom be, aminek segítségével sok időt spórolhatunk meg kétváltozós függvények szélsőértékhelyeinek meghatározásánál. Célom az volt, hogy az eljárást egy paraméterrel, magával a vizsgált függvénnyel lehessen meghívni, outputként pedig kapjuk meg a szélsőértékhelyeit és nyeregpontjait. Az eljárást Maple-ben lehet futtatni, miután betöltöttük a szükséges csomagokat. A könnyebb követhetőség érdekében készítettem egy egyoldalas folyamatábrát, amit a most következő kód után találunk. A magyarázat minden esetben a már leírt sorokra vonatkozik.

```
with(RealDomain):
with(LinearAlgebra):
with(Student[VectorCalculus]):
```

Ezzel a három sorral az eljárás futtatásához szükséges programcsomagokat töltöttem be. Az első azért kellett, hogy a *solve* parancs – ami egyenletrendszereket old meg – csak valós értékeket adjon vissza. A másik kettő pedig a Hesse-mátrix előállításához, definitségének megállapításához szükséges.

```
szels:=proc(f)
local jel, jelsz, i, a, b, j, p1, p2, c, d, H, cc, dd, van;
print(f);
```

Itt az első sorral elkezdtem az eljárást és egyben a *szels* nevet adtam neki. Az első sorból még annyi derül ki, hogy az eljárást a – *szels(függvény);* – sor beírásával hívhatjuk meg. A második sorban deklaráltam néhány változót annak érdekében, hogy a későbbiekben kevesebbet kelljen gépelni, remélhetőleg a program is áttekinthetőbbé vált tőle. Az utolsó sorban kiírtam a képernyőre a bevitt függvényt, ez lesz az output első sora, tehát az eljárás lefutása után ez lesz az első sor, amit visszakapunk.

```
jel:=convert(convert(solve({diff(f,x)=0, diff(f,y)=0},[x,y]),'set'),'list'):
jelsz:=nops(jel):
print();
```

A *jel* nevű változóba betöltöttem az f fügvény lehetséges szélsőértékhelyeinek listáját. Ezeknek a meghatározásában a *solve* parancs segített, ami egy listát ad azokról a pontokról, ahol mindkét parciális derivált egyenlő nullával. Mivel volt olyan függvény, ahol egy pontot többször is felsorolt, a *convert* parancs kétszeri alkalmazásával a listát halmazzá, majd a halmazt ismét listává konvertáltam. Így kihullottak a többszörös felsorolások és megmaradt a lista, amire szükségem lesz a továbbiakban. A második sorban a $jelsz$ változó értéke egyenlő lesz a lehetséges szélsőértékhelyek számával, azaz a *jel* lista elemeinek számával, majd hagyunk egy üres sort.

```
for i from 1 to jelsz
do
  a:=subs(jel[i][1],x):
  b:=subs(jel[i][2],y):
```

Itt elkezdődik egy ciklus, ami sorra veszi a *jel* lista elemeit, amik számpárok. Az a változóba tárolom az éppen vizsgált pont első koordinátáját, a b -be pedig a másodikat. Amikor a későbbiekben visszatérünk a ciklus elejére, akkor újraindul, ha van még nem vizsgált elem a listában, egyébként vége az eljárásnak.

```
  H:=subs(x=a,y=b,Hessian(f,[x,y]));

  if IsDefinite(H,'query' = 'positive_definite')
  then (print("Szigorú lokális minimum:"),print(jel[i]),print(),print(H))
  else

    if IsDefinite(H,'query' = 'negative_definite')
    then (print("Szigorú lokális maximum:"),print(jel[i]),print(),print(H))
    else

      if IsDefinite(H,'query' = 'indefinite')
      then (print("Nincs lokális szélsőérték:"),print(jel[i]),print(),print(H))
      else
```

Az első sorban előállítjuk és a H változóba tároljuk az $f(a,b)$-hez tartozó Hesse-mátrixot. Majd következik egy *ha–akkor–egyébként* sorozat, ami először azt vizsgálja, hogy a H mátrix pozitív definit vagy sem. Ha igen, akkor kiírja, hogy a vizsgált pontban szigorú lokális minimumot talált, majd visszatér a ciklus elejére. Ha a H mátrix nem pozitív definit, akkor jön a következő sor, ahol szintén vagy kiírja, hogy talált szélsőértékhelyet és visszatér a ciklus elejére, vagy jön a következő sor. A harmadik feltételnél azt dönti el a program, hogy a H mátrix indefinit-e. Itt vagy kiírja, hogy nincs szélsőértékhely, majd ciklus eleje vagy jön a következő sor.

```
if (subs(x=a,y=b,diff(diff(f,x),x))<>0 and
    subs(x=a,y=b,diff(diff(f,y),y))<>0)
then
  if IsDefinite(H,'query' = 'positive_semidefinite')
  then (print("Lokális minimum:"),print(jel[i]),print(),print(H))
  else
    if IsDefinite(H,'query' = 'negative_semidefinite')
    then (print("Lokális maximum:"),print(jel[i]),print(),print(H))
    end if
  end if
else
```

Az első feltétel szerint, ha igaz, hogy $f''_{xx}(a,b) \neq 0$ és $f''_{yy}(a,b) \neq 0$, akkor a Hesse-mátrix alapján dönt, hogy az adott pontban lokális minimum vagy lokális maximum van, majd visszatér a ciklus elejére. Ha $f''_{xx}(a,b) = 0$ vagy $f''_{yy}(a,b) = 0$, akkor jön a következő sor.

```
c:=diff(diff(f,x),x);
d:=diff(diff(f,y),y);
p1:=0;
p2:=0;
```

A c változóba tároljuk az $f''_{xx}(x,y)$ függvényt, a d -be pedig az $f''_{yy}(x,y)$ függvényt. A $p1$ és $p2$ változók értékét nullára állítjuk.

```
if c<>0 then for p1 from 0 while subs(x=a,y=b,c)=0 do c:=diff(c,x) end do end if;
if d<>0 then for p2 from 0 while subs(x=a,y=b,d)=0 do d:=diff(d,y) end do end if;
```

Ha $c \neq 0$, akkor megszámoljuk és $p1$ -be tároljuk, hogy a $c(a,b)$ -t hányszor kellett x szerint parciálisan deriválni ahhoz, hogy az értéke ne nulla legyen. Ha $d \neq 0$, akkor megszámoljuk és $p2$ -be tároljuk, hogy a $d(a,b)$ -t hányszor kellett y szerint parciálisan deriválni ahhoz, hogy az értéke ne nulla legyen.

```
cc:=subs(x=a,y=b,c);
dd:=subs(x=a,y=b,d);
van:=modp(p1,2)=0 and modp(p2,2)=0;
```

Az esetleges parciális deriválások után a cc változóba tároljuk $c(a,b)$ értékét, a dd változóba pedig $d(a,b)$ értékét. Létrehozunk egy van nevű logikai változót, ami akkor igaz, ha $p1 = p2 = 0 \ mod \ 2$ igaz.

```
          if (cc>0 and dd>0 and van)
          then (print("Szigorú lokális minimum:"),print(jel[i]),print())
          else
            if (cc<0 and dd<0 and van)
            then (print("Szigorú lokális maximum:"),print(jel[i]),print())
            else
              if (cc>=0 and dd>=0 and van)
              then (print("Lokális minimum:"),print(jel[i]),print())
              else
                if (cc<=0 and dd<=0 and van)
                then (print("Lokális maximum:"),print(jel[i]),print())
                else
                  (print("Nincs lokális szélsőérték:"),print(jel[i]),print())
```

Itt *cc*,*dd* és a *van* változók értéke alapján döntünk, hogy találtunk-e (szigorú) lokális szélsőértékhelyet vagy sem. Az eredményt kiírjuk, majd visszatérünk a ciklus elejére.

```
                end if
              end if
            end if
          end if
        end if
      end if
    end if;
  end if;
end do
end proc:
```

Ezek az elágazások lezárásai és az eljárás utolsó sora. Ha lefutott, akkor a képernyőre kiírva kapjuk a megtalált lokális szélsőértékhelyeket és nyeregpontokat. A program nem lett tökéletes, mert amikor a tanult tételek alapján nem tudunk dönteni, akkor néha a program sem tud dönteni vagy esetleg rossz megoldást kaphatunk.


START
Input: f(x,y)
STOP
Nem
Jel := lehetséges szélsőértékhelyek listája;
Jelsz := jelöltek száma;
i := 1;
a := jel[i,1];
b := jel[i,2];
H := Hesse(f, (a,b));
Igen
i <= jelsz;
cc > 0 és dd > 0 és van=igaz
Igen
H pozitív definit
Igen
Output:
f(x,y);
Szigorú lokális minimum;
(a,b);
Nem
cc => 0 és dd => 0 és van=igaz
Igen
Nem
H negatív definit
Igen
Output:
f(x,y);
Szigorú lokális maximum;
(a,b);
Nem
Nem
H indefinit
Igen
Output:
f(x,y);
Nyeregpont;
(a,b);
i:=i+1;
cc < 0 és dd < 0 és van=igaz
Igen
Nem
$f''_{xx}(a,b) \neq 0 \wedge f''_{yy}(a,b) \neq 0$
Igen
H pozitív szemidefinit
Igen
Output:
f(x,y);
Lokális minimum;
(a,b);
Nem
cc <= 0 és dd <= 0 és van=igaz
Igen
Nem
Nem
$c := f''_{xx}$;
$d := f''_{yy}$;
$p1 := p2 := 0$;
H negatív szemidefinit
Igen
Output:
f(x,y);
Lokális maximum;
(a,b);
$c \neq 0$;
Igen
Amíg c(a,b)=0:
$c := c'_x$;
$p1 := p1 + 1$;
Nem
$d \neq 0$;
Igen
Amíg d(a,b)=0:
$d := d'_y$;
$p2 := p2 + 1$;
cc:=c(a,b);
dd:=d(a,b);
van:=Igaz, ha p1=p2=0 mod 2;
Nem

Összefoglalás

Szakdolgozatomban ismertettem az $\mathbb{R}^p \to \mathbb{R}$ függvényekhez tartozó azokat a fontosabb tételeket, definíciókat, melyek megértéséhez és alkalmazásához segítségünkre szolgálhat egy számítógépes program, jelen esetben a Maple. Úgy gondolom, hogy ahol lehetett, ott informatív ábrákat készítettem az egyes fejezetekhez, feladatokhoz, amik talán segítenek feldolgozni az anyagrészt. Az ábrák mellett azt is leírtam, hogy ezeket hogyan lehet Maple segítségével előállítani. Írtam egy szélsőértékhely kereső programot, ami ugyan nem lett tökéletes, de a tankönyvekben szereplő függvények túlnyomó részénél jó megoldást kapunk a használatával, ami jelentősen felgyorsítja a feladatok megoldását vagy ellenőrzését. Néhány feladat keretében bemutattam, hogy a tanult tételek és módszerek nem minden esetben működnek, ekkor milyen lehetőségek vannak még a szélsőértékhelyek azonosítására.

Irodalomjegyzék

[1] Laczkovich Miklós, T. Sós Vera, *Analízis II. kötet*, Nemzeti Tankönyvkiadó, Budapest, 2007.

[2] George B. Thomas, *Thomas-féle Kalkulus III. kötet*, Typotex, Budapest, 2007.

[3] Keleti Tamás, *Órai jegyzet*, 2009-2010-2.